

Amplitúdový modulátor s AD633

Jakub Janek

2. mája 2022

Obsah

1	Zadanie experimentov	3
2	Charakteristika modelovanej súčiastky	4
3	Zapojenie amplitúdového modulátora	5
4	Experiment d)	6
5	Experiment e)	8

1 Zadanie experimentov

- a) Získajte z internetu alebo iným spôsobom model SPICE a katalógový list ("datasheet") analógovej násobičky AD633 fy Analog Devices.
- b) Súčiastku stručne popíšte - funkcie, parametre, zoznam možných aplikácií.
- c) S integrovaným obvodom zostavte lineárny amplitúdový modulátor podľa doporučeného zapojenia uvedeného v katalógovom liste. Popíšte čo najdetailnejšie funkciu zapojenia.
- d) Modulujte nosný signál o frekvencii 50 kHz sínusovým napätím 1 kHz. Zistite a popíšte, ako sa dá nastaviť hĺbka modulácie (vyjdete zo vzorca v katalógovom liste). Zobrazte spektrum amplitúdovo modulovaného signálu a stručne popíšte.
- e) Upravte zapojenie tak, aby ste dosiahli modulácie DSB-SC (potlačená nosná). Zobrazte časové priebehy výstupného signálu a príslušné amplitúdové spektrá.
- f) Prevedte ďalší experiment podľa vlastného uváženia.
- g) Odovzdajte protokol s týmito údajmi:
 - hlavička so zadáním, menom autora a dátumom odovzdania,
 - charakteristika modelovanej súčiastky podľa bodu b),
 - SPICE model prvku spolu s údajmi o spôsobe jeho získania,
 - schémy z katalógového listu s odkazom na tento list a popis testovaného zapojenia podľa bodu c),
 - výpisy vstupných súborov SPICE (prípadne výstupy zo schématického editoru) pre dané simulácie podľa bodov d), e), f), s popisom simulácii s ich výsledkami (relevantné výpisy zo vstupných súborom a výstupov z PROBE),
 - slovný komentár k získaným výsledkom,
 - prehľad literatúry a všetkých ďalších použitých zdrojov informácií vrátane internetovej adresy odkiaľ bol stiahnutý katalógový list a ďalšie údaje o súčiastke,
 - výpis vlastnej knižnice, v ktorej bude model integrovaného obvodu, prípadne modely ďalších súčiastiek, ktoré ste pre Váš projekt použili.

2 Charakteristika modelovanej súčiastky


AD633 je funkcionálne kompletná, štvorkvadrantová analógova násobička. Obsahuje diferenciálne vstupy X, Y a sumačný vstup s veľkou impedanciou (Z).

Parametre súčiastky:

- Vstupné napätie od 8 V až 18 V. Vnútorne napätie je generované Zenerovou diodou.
- $10\text{ M}\Omega$ vstupná impedancia.
- Pracovná teplota (J typ) - od 0°C až 70°C , -40°C až 85°C (A typ).

AD633 sa dá použiť na:


- Násobenie, delenie, druhú mocninu,
- modulácia, demodulácia, detekcia fáze,
- napätím ovládané zosilnenie/tlmenie/filtre.



Obr. 1: Vnútorne zapojenie AD633 [1]

3 Zapojenie amplitúdového modulátora

Amplitúdový modulátor je zapojenie v ktorom sa na nosný signál moduluje modulačný signál o nižšom kmitočte. Výstupom je signál frekvencie nosného signálu ktorý mení svoju amplitúdu podľa modulačného signálu.



Obr. 2: Upravené zapojenie AD633 ako lineárneho amplitúdového modulátora. [1]

Pri zapojení amplitúdového modulátoru s AD633, privádzame na AD633 symetrické napájanie $\pm 15\text{ V}$. Na napájanie sú taktiež pripojené blokovacie kondenzátory na zem o hodnote 100 nF , ktoré slúžia na odvedenie vysokofrekvenčného rušenia z napájania súčiastky. Na diferenčný vstup **X1**, **X2** je privedený vstup modulačného napätia. Tieto dva vstupy sú privedené na vstupný operačný zosilovač ktorého výstup je pripojený na násobičku spoločne s so vstupmi **Y1**, **Y2**, kde je následne ich súčin sčítaný so vstupom **Z** ktorý je taktiež pripojený k nosnej vlne. Výsledný súčet je prenesený na výstup **W** pomocou sledovača napätia.

$$W = \left(1 + \frac{E_M}{10V}\right) E_C \sin \omega t \quad (1)$$


Rovnica výstupného signálu **W**, je výsledkom spojenia analógových signálov E_M a E_n a následným sčítaním vstupného signálu E_n viď obrázok 2. Zapojenie a jeho simulácia prebiehala v simulačnom prostredí SPICE [2]. Netlist k tomuto zapojeniu sa nachádza v prílohe A.

4 Experiment d)

Amplitúdový modulátor zapojený na obrázku 2 moduluje výstupný signál **W** podľa vzorcu (1). Hĺbka modulácie je definovaná podľa vzorcu:


$$m = \frac{U_m}{U_n} \cdot 100 [\%] \quad (2)$$

Zo vzorcu (2) vieme, že zmena hĺbky je závislá na amplitúde nosného a modulačného signálu. Výstupný signál z AD633 **W** je taktiež na týchto amplitúdach závislý. Nastavenie hĺbky modulácie dochádza pri zmene týchto napätí.



Obr. 3: Spektrum amplitúdovej modulácie výstupného signálu **W** zapojenia na obrázku 2. [2]


Amplitúdové spektrum obsahuje tri harmonické zložky, kde nosný signál je o frekvencii 50 kHz a dve postranné zložky $f_n - f_m$ a $f_n + f_m$, teda 49 kHz a 51 kHz [2]. Výsledné spektrum odpovedá teórii o amplitúdovej modulácii.




Obr. 4: Výstup tranzientnej analýzy signálu **W**. [2]

5 Experiment e)


Úpravou pôvodného lineárneho amplitúdového modulátoru na obrázku 2 môžeme dosiahnuť aj modulácie DSB-SC (Double-sideband suppressed-carrier), odpojením nosného signálu od svorky **Z**. Po odpojení nosného signálu nedochádza k jeho pričítaniu. Výsledný signál neobsahuje nosnú frekvenciu ale len postranné pásma 49 kHz a 51 kHz. [2]



Obr. 5: Zapojenie AD633 ako DSB-SC modulátor.




Obr. 6: Frekvenčné spektrum výstupného signálu z DSB-SC modulátora. [2]



Obr. 7: Výsledný signál DSB-SC (potlačená nosná) vychádzajúci zo zapojenia na obrázku 5. [2]

6 Experiment f)

Ako posledný experiment sme zvolili, zistenie maximálneho možného napätia na výstupe. Na vstup **Y1** sme priviedli nosný signál s amplitúdou $U_n = 10\text{ V}$ a frekvenciou 50 kHz a na vstup **X1** modulačný signál s amplitúdou $U_m = 7\text{ V}$ s frekvenciou 1 kHz. Podľa rovnice (1) by výsledný maximálny modul výstupného signálu mal byť $U_{Wmax} = 17\text{ V}$. Toto napätie je ale ovplyvnené napájaním AD633, ktoré napájame symetrickým napájaním $\pm 15\text{ V}$. Po simulácii sme namerali maximálne napätie $U_{Wmax} = 11,31\text{ V}$ [2]. Toto napätie sa približuje hodnote uvedenej v katalógovom liste $\pm 11\text{ V}$ [1].



Obr. 8: Výsledný signál AM modulátora pri zvýšenom napätí nosnej vlny.

7 Záver

V tejto práci sme sa venovali analógovej násobičke AD633 v zapojení lineárneho amplitúdového modulátora a modulátora DSB-SC. Zapojenie uvedené v katalógovom liste [1] nebolo možné nasimulovať v programe kompatibilnom s PSPICE. Upravené zapojenie vid' obrázok 2 už dokázalo vytvoriť amplitúdovú moduláciu. Vo vlastnom experimente sme sa zaoberali maximálnym výstupným napätím pri napájaní $\pm 15\text{ V}$, výsledkom bolo blížiacie sa napätie uvedené v katalógovom liste $\pm 11\text{ V}$

Použité zdroje

1. ANALOG DEVICES. *Low Cost Analog Multiplier* [online]. [cit. 2022-04-30]. Dostupné z : <https://www.analog.com/media/en/technical-documentation/data-sheets/AD633.pdf>.
2. NGSPICE. *ngspice mixed level/mixed-signal circuit simulator* [http://ngspice.sourceforge.net]. 2022. Ver. 36. software.
3. ANALOG DEVICES. *AD633 SPICE Macro Model Analog Multiplier Rev. A, 12/93* [online]. [cit. 2022-04-30]. Dostupné z : <https://www.analog.com/media/en/simulation-models/spice-models/ad633.cir>.

A Amplitúdový modulátor v SPICE

```
Amplitudovy modulator s AD633
*
.inc AD633.lib
*
.param nosna_f=50k
.param modulovana_f=1k
*
*-- X1 X2 Y1 Y2 VNEG Z W VPOS AD633
XAM X1 0 Y1 0 VNEG Y1 W VPOS ad633; AD633

* modulovaný a nosný vstupný signál
Vm X1 0 AC 1 0 SIN(0 7 {modulovana_f})
Vn Y1 0 AC 1 0 SIN(0 5 {nosna_f})

* napájanie
Vpos VPOS 0 15V
Vneg VNEG 0 -15V

C1 VPOS 0 0.1u
C2 VNEG 0 0.1u
*
.control
probe V(W,0) V(X1,0) V(Y1,0)

tran 100n 2m
gnuplot plot/w-am v(W)

*tran 1u 100m
*linearize v(W)
*set specwindow=gaussian
*fft v(W)
*gnuplot plot/fftam mag(V(W))
.endc
.end
```

B DSB-SC modulátor v SPICE

```
DSB-SC AD633
*
.inc AD633.lib
*
.param nosna_f=50k
.param modulovana_f=1k
*
*-- X1 X2 Y1 Y2 VNEG Z W VPOS AD633
XAM X1 0 Y1 0 VNEG 0 W VPOS ad633; AD633

* modulovaný a nosný vstupný signál
Vm X1 0 AC 1 0 SIN(0 7 {modulovana_f})
Vn Y1 0 AC 1 0 SIN(0 5 {nosna_f})

* napájanie
Vpos VPOS 0 15V
Vneg VNEG 0 -15V

C1 VPOS 0 0.1u
C2 VNEG 0 0.1u
*
.control
probe V(W,0) V(X1,0) V(Y1,0)

tran 100n 2m
gnuplot plot/w-am v(W)

*tran 1u 100m
*linearize v(W)
*set specwindow=gaussian
*fft v(W)
*gnuplot plot/fftam mag(V(W))
.endc
.end
```

C Knižnica AD633 [3]

```

* AD633 Analog Multiplier Macro Model
* Description: Amplifier
* Generic Desc: Bipolar, Multiplier, 4 Quadrant
* Developed by: AAG/PMI
* Revision History: 09/15/2014 - Input resistors nodes connection change (RX1A & RY1A)
* 08/10/2012 - Updated to new header style
* 1.0 (12/1993)
* Copyright 1993, 2012 by Analog Devices
*
* Refer to http://www.analog.com/Analog\_Root/static/techSupport/designTools/spiceModels/license/sp
* indicates your acceptance of the terms and provisions in the License Statement.
*
* BEGIN Notes:
*
* Not Modeled:
*
* Parameters modeled include:
*
* END Notes
*
* Node assignments
*
*      X1
*      |  X2
*      |  |  Y1
*      |  |  |  Y2
*      |  |  |  |  VNEG
*      |  |  |  |  |  Z
*      |  |  |  |  |  |  W
*      |  |  |  |  |  |  |  VPOS
*      |  |  |  |  |  |  |  |
.SUBCKT AD633 1 2 3 4 5 6 7 8
*
EREF 100 0 POLY(2) 8 0 5 0 (0,0.5,0.5)
*
* X-INPUT STAGE & POLE AT 15 MHz
*
IBX1 1 0 DC 8E-7
IBX2 2 0 DC 8E-7
EOSX 10 1 POLY(1) (16,100) (5E-3,1)
RX1A 1 11 5E6
RX1B 11 2 5E6
*
GX 100 12 10 2 1E-6
RX 12 100 1E6
CX 12 100 1.061E-14
VX1 8 13 DC 3.05
DX1 12 13 DX
VX2 14 5 DC 3.05
DX2 14 12 DX
*
* COMMON-MODE GAIN NETWORK WITH ZERO AT 560 Hz
*
ECMX 15 100 11 100 10
RCMX1 15 16 1E6
CCMX 15 16 2.8421E-10
RCMX2 16 100 1

```

```

*
* Y-INPUT STAGE & POLE AT 15 MHz
*
IBY1 3 0 DC 8E-7
IBY2 4 0 DC 8E-7
EOSY 20 3 POLY(1) (26,100) (5E-3,1)
RY1A 3 21 5E6
RY1B 21 4 5E6
*
GY 100 22 20 4 1E-6
RY 22 100 1E6
CY 22 100 1.061E-14
VY1 8 23 DC 3.05
DY1 22 23 DX
VY2 24 5 DC 3.05
DY2 24 22 DX
*
* COMMON-MODE GAIN NETWORK WITH ZERO AT 560 Hz
*
ECMY 25 100 21 100 10
RCMY1 25 26 1E6
CCMY 25 26 2.8421E-10
RCMY2 26 100 1
*
* Z-INPUT STAGE & POLE AT 15 MHz
*
IBZ1 7 0 DC 8E-7
IBZ2 6 0 DC 8E-7
RZ1 7 6 10E6
*
GZ 100 32 7 6 1E-6
RZ2 32 100 1E6
CZ 32 100 1.061E-14
VZ1 8 33 DC 3.05
DZ1 32 33 DX
VZ2 34 5 DC 3.05
DZ2 34 33 DX
*
* 50-MHz MULTIPLIER CORE & SUMMER
*
GXY 100 40 POLY(2) (12,100) (22,100) (0,0,0,0,0.1E-6)
RXY 40 100 1E6
CXY 40 100 3.1831E-15
*
* OP AMP INPUT STAGE
*
VOOS 59 40 DC 5E-3
Q1 55 32 60 QX
Q2 56 59 61 QX
R1 8 55 3.1831E4
R2 60 54 3.1313E4
R3 8 56 3.1831E4
R4 61 54 3.1313E4
I1 54 5 1E-4
*
* GAIN STAGE & DOMINANT POLE AT 316.23 Hz
*
G1 100 62 55 56 3.141637E-5

```

```

R5 62 100 1.0066E8
C3 62 100 5E-12
V1 8 63 DC 4.3399
D1 62 63 DX
V2 64 5 DC 4.3399
D2 64 62 DX
*
* NEGATIVE ZERO AT 20 MHz
*
ENZ 65 100 62 100 1E6
RNZ1 65 66 1
FNZ 65 66 VNC -1
RNZ2 66 100 1E-6
ENC 67 0 65 66 1
CNZ 67 68 7.9577E-9
VNC 68 0 DC 0
*
* POLE AT 4 MHz
*
G2 100 69 66 100 1E-6
R6 69 100 1E6
C2 69 100 3.9789E-14
*
* OP AMP OUTPUT STAGE
*
FSY 8 5 POLY(2) VZC1 VZC2 (2.8286E-3,1,1)
RDC 8 5 28E3
GZC 100 73 72 69 11.623E-3
VZC1 74 100 DC 0
DZC1 73 74 DX
VZC2 100 75 DC 0
DZC2 75 73 DX
VSC1 70 72 0.695
DSC1 69 70 DX
VSC2 72 71 0.695
DSC2 71 69 DX
G01 72 8 8 69 11.623E-3
R01 8 72 86
G02 5 72 69 5 11.623E-3
R02 72 5 86
LO 72 7 1E-7
*
* MODELS USED
*
.MODEL QX NPN(BF=1E4)
.MODEL DX D(IS=1E-15)
.ENDS AD633

```